

Pensieve header: Testing notebook for Scatter and Glow in OneCo. Continues pensieve://2015-12/, continued pensieve://2016-02/.

In the U(T)U(H) conventions.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2016-01"];
<< SnG.m
```

Bases

```
UUBasis[{1, 2}, {3, 4}, f]
{UU[β[f1[b1, b2]]], UU[a[f2[b1, b2], 1, 3]], UU[δa[f3[b1, b2], 1, 3]],
UU[a[f4[b1, b2], 1, 4]], UU[δa[f5[b1, b2], 1, 4]], UU[a[f6[b1, b2], 2, 3]],
UU[δa[f7[b1, b2], 2, 3]], UU[a[f8[b1, b2], 2, 4]], UU[δa[f9[b1, b2], 2, 4]],
UU[δβ[f10[b1, b2]]], UU[c[f11[b1, b2], 3]], UU[c[f12[b1, b2], 4]],
UU[ca[f13[b1, b2], 3, 1, 3]], UU[ca[f14[b1, b2], 3, 1, 4]],
UU[ca[f15[b1, b2], 3, 2, 3]], UU[ca[f16[b1, b2], 3, 2, 4]],
UU[ca[f17[b1, b2], 4, 1, 3]], UU[ca[f18[b1, b2], 4, 1, 4]], UU[ca[f19[b1, b2], 4, 2, 3]],
UU[ca[f20[b1, b2], 4, 2, 4]], UU[δaa[f21[b1, b2], 1, 3, 1, 3]],
UU[δaa[f22[b1, b2], 1, 3, 1, 4]], UU[δaa[f23[b1, b2], 1, 4, 1, 4]],
UU[δaa[f24[b1, b2], 1, 3, 2, 3]], UU[δaa[f25[b1, b2], 1, 3, 2, 4]],
UU[δaa[f26[b1, b2], 1, 4, 2, 4]], UU[δaa[f27[b1, b2], 2, 3, 2, 3]],
UU[δaa[f28[b1, b2], 2, 3, 2, 4]], UU[δaa[f29[b1, b2], 2, 4, 2, 4]]}
```

Meta-associativity for tm

```
((# // tm[1, 2, 1] // tm[1, 3, 1]) == (# // tm[2, 3, 2] // tm[1, 2, 1])) & /@
UUBasis[{1, 2, 3, 4}, {5, 6}, f] // Union
{True}
```

Meta-associativity for hm

```
((# // hm[1, 2, 1] // hm[1, 3, 1]) == (# // hm[2, 3, 2] // hm[1, 2, 1])) & /@
UUBasis[{5, 6}, {1, 2, 3, 4}, f] // Union
{True}
```

Compatibility between tm and hts

```
(((# // tm[1, 2, 1] // hts[4, 1]) ==
(# // hts[4, 1] // hts[4, 2] // tm[1, 2, 1])) & /@
UUBasis[{1, 2, 3}, {4, 5}, f] // Union
{True}
```

Compatibility between hm and hts

```
((# // hm[1, 2, 1] // hts[1, 4]) ==
  (# // hts[2, 4] // hts[1, 4] // hm[1, 2, 1])) & /@
  UUBasis[{4, 5}, {1, 2, 3}, f] // Union
{True}
```

Meta-associativity for dm

```
((# // dm[1, 2, 1] // dm[1, 3, 1]) == (# // dm[2, 3, 2] // dm[1, 2, 1])) & /@
  UUBasis[4, f] // Union
{True}
```

Meta-Jacobi for tb

```
tbAS[u_, v_] := tb[0][u, v] + tb[0][v, u];
DeleteCases[
  Flatten[Outer[
    {##} → tbAS[##] &,
    UUBasis[{0, 1}, {1, 2}, f], UUBasis[{0, 2}, {3, 4}, f]
  ]],
  _ → UU[0]
]
{}

tbJacobi[u_, v_, w_] :=
  tb[0][u, tb[0][v, w]] + tb[0][v, tb[0][w, u]] + tb[0][w, tb[0][u, v]];
DeleteCases[
  Flatten[Outer[
    {##} → tbJacobi[##] &,
    UUBasis[{0, 1}, {1, 2}, f], UUBasis[{0, 2}, {3, 4}, g], UUBasis[{0, 3}, {5, 6}, h]
  ]],
  _ → UU[0]
]
{}

```

Meta-Jacobi for hb

```

hbAS[u_, v_] := hb[0][u, v] + hb[0][v, u];
DeleteCases[
  Flatten[Outer[
    {##} → hbAS[##] &,
    UUBasis[{1, 2}, {0, 1}, f], UUBasis[{3, 4}, {0, 2}, f]
  ]],
  _ → UU[0]
]
{}

hbJacobi[u_, v_, w_] :=
  hb[0][u, hb[0][v, w]] + hb[0][v, hb[0][w, u]] + hb[0][w, hb[0][u, v]];
DeleteCases[
  Flatten[Outer[
    {##} → hbJacobi[##] &,
    UUBasis[{1, 2}, {0, 1}, f], UUBasis[{3, 4}, {0, 2}, g], UUBasis[{5, 6}, {0, 3}, h]
  ]],
  _ → UU[0]
]
{}

```

ct (contract) poperties

```

Outer[
  Equal[
    ct[#1 // tm[1, 2, 1], #2],
    ct[#1, #2] // tm[1, 2, 1]
  ] &,
  UUBasis[{1, 2, 3}, {0, 1}, f], UUBasis[{0, 4}, {2, 3}, g]
] // Flatten // Union
{True}

```

```
Outer[
  {#1, #2} → Equal[
    ct[#1 // hm[1, 2, 1], #2],
    ct[#1, #2] // hm[1, 2, 1]
  ] &,
  UUBasis[{1, 2}, {0, 1, 2, 3}, f], UUBasis[{0, 4}, {4, 5}, g]
] // Flatten // DeleteCases[#, _ → True] &
{}
```

```
Outer[
  {#1, #2} → Equal[
    ct[#1, #2 // hm[2, 3, 2]],
    ct[#1, #2] // hm[2, 3, 2]
  ] &,
  UUBasis[{1, 2}, {0, 1}, f], UUBasis[{0, 3}, {2, 3, 4}, g]
] // Flatten // DeleteCases[#, _ → True] &
{}
```

```
Outer[
  {#1, #2} → Equal[
    ct[#1, #2 // tm[3, 4, 3]],
    ct[#1, #2] // tm[3, 4, 3]
  ] &,
  UUBasis[{1, 2}, {0, 1}, f], UUBasis[{0, 3, 4, 5}, {2, 3}, g]
] // Flatten // DeleteCases[#, _ → True] &
{}
```